Website URL

Skype

AIM

MSN

ICQ

Yahoo

Jabber

Location

Interests

Mods Worked On

We have found out in other threads that some "awareness" of things presented in mod-added content can be scripted directly into your mod without any troubles. For example, Aran can reference any existing DV, so !Inparty("mynpc") checks will not fail on install, and the strings referenced will be available if the mod-added NPC shows up in-game later on, regardless of install order.
This works with globals, too, so in other threads progress to certain points in other mods allows "side talk" - not directly bantering with an NPC or using an item, but acknowledging their presence, either by use of <PLAYER3> tokens in dialog, or with direct checks for globals and DVs. Any way we cut this, we are still talking about having Aran (or your NPC) having some of his talks and paths refer to outside events created by other mods, and being able to install that content as part of Aran's mod. The presence of the global variable or DV being active in a game then "activates" that content passively - no need for messing about with other folk's files, etc.
We have taken this a step further with looking at player responses by mod-added kit, in the thread regarding "what happens when..."; here, we found that when Divine Remix was not installed and we wanted to reference the kits added, we had a problem. Compiling a reference to a non-pre-existing kit left the engine with only one choice, "0". Which caused all sorts of problems if DR was not installed, and even worse had no "repair" effect on installed files - unlike GTIMES.IDS or other "engine lookup by handle" things, KIT.IDS references have to be built *after* the kits have been placed on an install, because of the way the files are created and referenced.
We get around that by creating an independent "patch", ARAN_AWARE, which so far just swaps "dummy" variables that will compile at install but never be triggered (a very old idea in terms of shutting down content, the equivalent of adding False() to a trigger so a dialog will never play, except reversing a False() is a PITA, whereas setting a never-reached variable to 1 is a much more compatibility - friendly way of coding it - a SetGlobal, and the original block plays.) with the wanted content using good old fashioned REPLACE_TEXTUALLY operation:
/* HEARTWARDER OF SUNE - Sune Firehair, Lady Firehair : Global("c-heartwarder","GLOBAL",1) */
ACTION_IF FILE_EXISTS_IN_GAME nmsune.2da BEGIN
ACTION_FOR_EACH ~dialog_file~ IN ~c-aranj.dlg~
~c-aranp.dlg~
~c-arn25j.dlg~
~c-arn25a.dlg~ BEGIN
ACTION_IF FILE_EXISTS_IN_GAME ~%dialog_file%~ THEN BEGIN
COPY_EXISTING ~%dialog_file%~ ~override~
DECOMPILE_DLG_TO_D
REPLACE_TEXTUALLY ~Global("c-heartwarder","GLOBAL",1)~ ~Kit(Player1,NMSUNE) OR(6) Class(Player1,CLERIC) Class(Player1,FIGHTER_CLERIC) Class(Player1,CLERIC_MAGE) Class(Player1,CLERIC_THIEF) Class(Player1,FIGHTER_MAGE_CLERIC) Class(Player1,CLERIC_RANGER)~
REPLACE_TEXTUALLY ~Global("c-aranheartwarder","GLOBAL",1)~ ~Kit("c-aran",NMSUNE) OR(6) Class("c-aran",CLERIC) Class("c-aran",FIGHTER_CLERIC) Class("c-aran",CLERIC_MAGE) Class("c-aran",CLERIC_THIEF) Class("c-aran",FIGHTER_MAGE_CLERIC) Class("c-aran",CLERIC_RANGER)~
COMPILE_D_TO_DLG
BUT_ONLY_IF_IT_CHANGES
END
END
END
So far so good. We know that using this after Divine Remix is installed means that the correct reference gets swapped into Aran's files. If a player installs the "patch" mod, the new content is activated. If they don't their game still works, and there is no problem.
So, what else would need this kind of handling?

One of my favorite NPCs - all these are by me, but hopefully Bri or Lord Ernie will wander by and make changes or suggestions.
File: aranw_aware\aranaware_tashia_SoAbanters.d
/* SoA Banters: Tashia<>Aran : High Spirits : cmorgan */
CHAIN IF ~Global("c-tashia1aran","GLOBAL",0)
InParty(Myself) !StateCheck(Myself,CD_STATE_NOTVALID)
InParty("c-aran") InMyArea("c-aran") !StateCheck("c-aran",CD_STATE_NOTVALID)
CombatCounter(0) !See([ENEMY])~
THEN BTASHIA tashia1arantalk
~[TASHIA] Rather gloomy, aren't you?~
DO ~SetGlobal("c-tashia1aran","GLOBAL",1)~
== C-ARANB ~[ARAN] Aye, then, lass, I think you have a mite bit o' th' truth there. Down just a little, I think.~
== BTASHIA ~[TASHIA] So, how about a riddle?~
== C-ARANB ~[ARAN] No, not now, as I got naught for spirited talk. You seem a mite down yourself, there. You got an interestin' familiar, some tough magic, an' you be travelin' with us. What's th' matter wi' you, to bring you so far down?~
== BTASHIA ~[TASHIA] Well, that's really no one's concern but mine. Personal matters... you understand, I'm sure.~
== C-ARANB ~[ARAN] Sure. I understand. Here, lass... have a nip from this flask. It may not solve anythin', but it will warm up your insides.~
== BTASHIA ~[TASHIA] I would think so... I can smell it from here!~
== C-ARANB ~[ARAN] Sometimes a nip o' somthin' what can kick a lad or lass will do wonders. Spirits for to raise your spirits, eh?~
== BTASHIA ~[TASHIA] I don't know...~
== C-ARANB ~[ARAN] Well, suit yourself. Th' offer's always available.~
EXIT
[code]
And now for a little moe fun - let's play with the blanket idea that Tashia has between Jaheira and Aerie, only throwing a slightly different light on thins - a riff in the key of flirt, if Tashia and Aerie are in the party with Aran:
[code]
APPEND C-ARANB
/* SoA Banters: Aran<>Tashia : no romance with PC with either Aran or Tashia, and even better if you have both Tashia and Aerie in the party... : cmorgan */
IF ~!Global("TashiaRomanceActive","GLOBAL",2) !Global("c-aranrom","GLOBAL",2) Global("c-aran1tashia","GLOBAL",0)
InParty(Myself) !StateCheck(Myself,CD_STATE_NOTVALID)
InParty("tashia") InMyArea("tashia") !StateCheck("tashia",CD_STATE_NOTVALID)
CombatCounter(0) !See([ENEMY])~ THEN BEGIN c-aran1tashiatalk
SAY ~[ARAN] Oh, nice an' soft... just th' right sort o' thing. I'll just be packin' this up, an'...~
IF ~~ THEN DO ~SetGlobal("c-aran1tashia","GLOBAL",1)~ EXTERN BTASHIA c-aran1tashiastart
END
IF ~~ c-aran1tashiafour
SAY ~[ARAN] Now, I doubt that would end so well for me. I done learned a long time ago that pleasin' a woman be a full time occupation, an' splittin' forces leads to disaster.~
IF ~~ THEN EXTERN BTASHIA c-aran1tashiaseven
END
IF ~~ c-aran1tashiafive
SAY ~[ARAN] Now, while that be a right entertainin' thought, I was actually talkin' to you, Tashia.~
IF ~~ EXTERN BTASHIA c-aran1tashiaseven
END
IF ~~ c-aran1tashiasix
SAY ~[ARAN] Now, while that be a right entertainin' thought, I was actually talkin' to you, Aerie.~
IF ~~ EXTERN BAERIE c-aran1tashiaeight
END
END
CHAIN C-ARANB c-aran1tashiaone
~[ARAN] Well, I might be in a right fine position to share it wi' you...~
== BTASHIA ~[TASHIA] Are you suggesting what I think you are suggesting?~
== C-ARANB ~[ARAN] Now, you be a woman o' th' world, an' a beautiful one at that. Why should I not ask?~
END
IF ~~ THEN EXTERN BTASHIA c-aran1tashiaseven
CHAIN BAERIE c-aran1tashiatwo
~[AERIE] No, that is mine! How did we get these all mixed up?~
== C-ARANB ~[ARAN] Well, I might be in a right fine position to share it wi' you...~
== BTASHIA ~[TASHIA] Are you suggesting what I think you are suggesting?~
== BAERIE ~[AERIE] Wh- what is he suggesting?~
== C-ARANB ~[ARAN] Now, you be a woman o' th' world, an' a beautiful one at that. Why should I not ask?~
== BAERIE ~[AERIE] Aran, are you talking to me, or to her? Ask what?~
END
IF ~~ THEN EXTERN BTASHIA c-aran1tashiathree
APPEND BTASHIA
IF ~~ c-aran1tashiastart
SAY ~[TASHIA] Hey! That's my blanket you're taking!~
IF ~InParty("aerie") InMyArea("aerie") StateCheck("aerie",CD_STATE_NOTVALID)~ THEN EXTERN BAERIE c-aran1tashiatwo
IF ~OR(3) !InParty("aerie") !InMyArea("aerie") StateCheck("aerie",CD_STATE_NOTVALID)~ THEN EXTERN C-ARANB c-aran1tashiaone
END
IF ~~ c-aran1tashiathree
SAY ~[TASHIA] For all we know, he might be talking to us both.~
IF ~RandomNum(3,3)~ THEN EXTERN C-ARANB c-aran1tashiafour
IF ~RandomNum(3,2)~ THEN EXTERN C-ARANB c-aran1tashiafive
IF ~RandomNum(3,1)~ THEN EXTERN C-ARANB c-aran1tashiasix
END
END
CHAIN BTASHIA c-aran1tashiaseven
~[TASHIA] I see someone has a growing appreciation for the possibility of female companionship.~
== C-ARANB ~[ARAN] Do you be usin' some sort o' sorcerous ways to see right through these clothes o' mine?~
== BTASHIA ~[TASHIA] No, no... I just know the way men's minds work.~
== BTASHIA ~[TASHIA] Nice try, Aran, but I have my Peanelian to keep me warm.~
== C-ARANB ~[ARAN] I do admit he can do that right well, but I might have some more ways o' pleasin' you, you know.~
== BTASHIA ~[TASHIA] Oh, I don't know...~
== C-ARANB ~[ARAN] You can't be naught in th' way o' serious.~
== BTASHIA ~[TASHIA] Hehe... You should see your face!~
== C-ARANB ~[ARAN] You might think on it a bit, you know, all jokin' aside. You be a fine lookin' lass.~
EXIT
CHAIN BAERIE c-aran1tashiaeight
~[AERIE] I... Oh. You are... You are?~
== C-ARANB ~[ARAN] I have been admirin' you for some time now, Aerie. I wouldn't speak so bold if I hadn't seen your eyes come my direction more than a few times, eh?~
== BTASHIA ~[TASHIA] I see someone has a growing appreciation for the possibility of female companionship.~
== C-ARANB ~[ARAN] Do you be usin' some sort o' sorcerous ways to see right through these clothes o' mine?~
== BTASHIA ~[TASHIA] No, no... I just know the way men's minds work.~
== BAERIE ~[AERIE] You... you were talking to me.~
== BTASHIA ~[TASHIA] Now I can't decide which of you two is blushing brighter. Should I just leave you two lovebirds alone?~
== BAERIE ~[AERIE] NO! No... I mean, please don't go. I... I don't think Aran is suggesting...~
== BTASHIA ~[TASHIA] Believe me, he is.~
== C-ARANB ~[ARAN] Belive me, I am.~
== BAERIE ~[AERIE] Oh! Oh my!~
== BTASHIA ~[TASHIA] Was her grabbing the blanket and running off hiding her face in it part of your master plan, loverboy?~
== C-ARANB ~[ARAN] Sune's Sweet Smile, I thought I was makin' time right well. Damnation on my blighted tongue.~
== BTASHIA ~[TASHIA] Sometimes a girl does like attention. Just don't go overboard.~
EXIT
installed and tested.

A tough one to write for.
/* SoA Banters: Aran<>Tsujatha : What are you... : cmorgan */
CHAIN IF ~Global("c-aran1tsujatha","GLOBAL",0)
InParty(Myself) !StateCheck(Myself,CD_STATE_NOTVALID)
InParty("TSUJATH") InMyArea("TSUJATH") !StateCheck("TSUJATH",CD_STATE_NOTVALID)
CombatCounter(0) !See([ENEMY])~
THEN C-ARANB c-aran1tsujathatalk
~[ARAN] What in th' nine hells are you? I never did see a Tel'Quessirim so tall. An' you don't rightly look natural.~
DO ~SetGlobal("c-aran1tsujatha","GLOBAL",1)~
== BTSUJAT ~[TSJUATHA] I am a necromancer. I do manipulate death. And this serves me well. It serves <CHARNAME> also.~
== C-ARANB ~[ARAN] You blighted well know that be naught in th' way o' an answer. An' don't try that cold stare wi' me. I have seen th' worst o' death, an' I don't rightly scare easily.~
== BTSUJAT ~[TSJUATHA] Have you walked the planes, brought animation to the dead, and weilded the powers of a Prime?~
== C-ARANB ~[ARAN] Well, seein' as I don't rightly know what two out o' th' three o' those really mean, mayhap not.~
== BTSUJAT ~[TSJUATHA] You have much to learn, then, before you can truly fear me.~
== C-ARANB ~[ARAN] You are a strange one, Tsjuatha. I don't know if havin' you around be a blessin' or damnation for us all.~
== BTSUJAT ~[TSJUATHA] I have embraced damnation with open arms--and open eyes.~
== C-ARANB ~[ARAN] Well, I blighted well haven't.~
EXIT
/* SoA Banters: Tsujatha<>Aran : Pride and Prejudice : cmorgan */
CHAIN IF ~Global("c-tsujatha1aran","GLOBAL",0)
InParty(Myself) !StateCheck(Myself,CD_STATE_NOTVALID)
InParty("TSUJATH") InMyArea("TSUJATH") !StateCheck("TSUJATH",CD_STATE_NOTVALID)
CombatCounter(0) !See([ENEMY])~
THEN BTSUJAT c-tsujatha1arantalk
~[TSJUATHA] You are a sword for hire, and a commoner of this land, Master Whitehand?~
DO ~SetGlobal("c-tsujatha1aran","GLOBAL",1)~
== C-ARANB ~[ARAN] Sellsword, aye. Commoner, well... common enough. Mayhap there be a bit o' social distinction between us non-nobles around these parts, but th' adventurin' an' sellsword life pulls to th' bottom o' th' heap most times.~
== C-ARANB ~[ARAN] Put it this way; a merchant's son you can bring home to th' family. A sellsword, well... you have a nice little affair, some kissin' an' sparkin' in th' bedchamber. Then you smarten up an' go find some nice boy o' th' settlin' down type to actually marry.~
== C-ARANB ~[ARAN] What about you? Do they all be so dark an' nihilistic where you be from?~
== BTSUJAT ~[TSJUATHA] I was born the only son of an ancient and proud house, one of the noble houses amongst the Fair Elves. Before my second century I was Prince of the Golden Desert.~
== C-ARANB ~[ARAN] I have naught in th' way o' experience talkin' wi' no Prince, let alone some sandy gold-lovin' plane-travelein' one. But I have had plenty o' experience dealin' wi' braggarts an' those what think they are superior.~
== BTSUJAT ~[TSJUATHA] (A smirk crosses his face at Aran's quaint phrasing, but his eyes are serious as he responds.)~
== BTSUJAT IF ~GlobalGT("NaliaTsujathaTalk","GLOBAL",1)~ THEN ~[TSUJATHA] I believe that I have said this to Mistress Nalia...~
== BTSUJAT ~[TSJUATHA] It would be best for your health if you gave up these assumptions. I am of higher birth than you are, but it hurts your pride to think so.~
== C-ARANB ~[ARAN] Higher birth on account o' you bein' not from this plane, on account o' you bein' born an elf, on account o' your family's bloodline, or on account o' you bein' a wizard what studies death?~
== BTSUJAT ~[TSJUATHA] All four of these. And none of the four.~
== C-ARANB ~[ARAN] Th' thing is, you talk a right fine game. You act an' look colder than death, mincin' about superiour to us poor blighted inhabitants o' this backwater plane. But you stay here, desperately pickin' fights an' pushin' at others around you so's to see yourself in th' reflections o' their conversations. So you want to live, that be sure, no matter how much you play wi' words n' shadow.~
== BTSUJAT ~[TSJUATHA] I would not recommend mocking those whose power exceeds your own.~
== C-ARANB ~[ARAN] It were naught in th' way o' mockery. An' there you go again, bein' all dark an' threatenin'. There be comfort an' strength in bein' part o' a group o' "lesser powers". I might not touch you one on one. But get a few o' my mates together, an' death magic or no, we'd balance you out right fine.~
EXIT
This second one needs to be revisited. It is part of the way that Tsujatha is written that he is smarter, darker, wiser, and more angsty than anyone else in the party. He *always* wins, and other NPCs *always* admit he is right, albeit sometimes grudgingly. Even in the limited Crossmod Banter Project entries, NPCs discussing things with him are kind to him.
Do I like this? No. Do I care deeply? No - this is fiction, and they are both toys. Do people like this? Yes - he is downloaded and played even today. SO... I should probably keep the banters in character with Tsujatha's character, and make sure he has a suitably dark comeback.
I'll log it on the to-do list, after I have read more about him.

Now, with the Granddaddy of All Mods, Solaufein, there is an added difficulty. We can (and I have) create two fun banters, and install them.
file: aranw_aware\aranaware_solaufein_SoAbanters.d
/* SoA Banters: Aran<>Solaufein : cmorgan */
CHAIN IF ~Global("c-aran1solaufein","GLOBAL",0)
InParty(Myself) !StateCheck(Myself,CD_STATE_NOTVALID)
InParty("SOLA") InMyArea("SOLA") !StateCheck("SOLA",CD_STATE_NOTVALID)
CombatCounter(0) !See([ENEMY])~
THEN C-ARANB c-aran1solaufeintalk
~[ARAN] So I hear you be a "good" Drow.~
DO ~SetGlobal("c-aran1solaufein","GLOBAL",1)~
== SOLA ~[sOLAUFEIN] Yes, it would appear so.~
== C-ARANB ~[ARAN] Didn't know there were any such thing.~
== SOLA ~[sOLAUFEIN] I assure you, I am. I exist, and I am not alone.~
== C-ARANB ~[ARAN] Well, I suppose you might find a rose on a dungheap, but I'd be thinkin' it would be a right easier thing to find a "good" demon servin' tea at Hareshom's o' Waterdeep.~
== SOLA ~[sOLAUFEIN] Perhaps you are correct. But the followers of Eilistraee, Lady Silverhair, must be discreet if we are to guide our brothers and sisters to a lighter path.~
== C-ARANB ~[ARAN] This Ellyartrie...~
== SOLA ~[sOLAUFEIN] Eilistraee.~
== C-ARANB ~[ARAN] Look, do there be many followers o' Her particular brand o' Drowdom?~
== SOLA ~[sOLAUFEIN] Very few, I am afraid.~
== C-ARANB ~[ARAN] Actually, that makes me a mite bit happier. I'd rather not be rethinkin' every single Drow bastard I personally put down. Chances are, none o' them be ones where I should o' hesitated.~
== SOLA ~[sOLAUFEIN] Ah, my friend, for a brief moment I thought that you would see past your prejudice rather that rationalize your past actions. I will sing a prayer to Lady of the Dance.~
EXIT
/* SoA Banters: Solaufein<>Aran : cmorgan */
CHAIN IF ~Global("c-solaufein1aran","GLOBAL",0)
InParty(Myself) !StateCheck(Myself,CD_STATE_NOTVALID)
InParty("'c-aran") InMyArea("'c-aran") !StateCheck("'c-aran",CD_STATE_NOTVALID)
CombatCounter(0) !See([ENEMY])~
THEN SOLA c-solaufein1arantalk
~[sOLAUFEIN] Ah, my prejudiced fellow traveller - speech is a mirror of the soul. As a man speaks, so is he. I ask you now: who is Aran?~
DO ~SetGlobal("c-solaufein1aran","GLOBAL",1)~
== C-ARANB ~[ARAN] I don't rightly understand th' question. Wait... mayhap I do. Do you be sayin' th' way a man speaks be a good indication o' his personality, his habits, an' so forth?~
== SOLA ~[sOLAUFEIN] Why, yes. I think that is a fair assessment.~
== C-ARANB ~[ARAN] That be a right interestin' idea. But I don't think it applies to me. You want to know me, you look at how I write, an' ignore th' language.~
== SOLA ~[sOLAUFEIN] I do not think that will be a fair picture. With your scribing, you have time to think, to craft your ideas. In your speech, there is an immediacy that shows a raw truth.~
== C-ARANB ~[ARAN] Mayhap. Or mayhap we have to agree to disagree, eh?~
== SOLA ~[sOLAUFEIN] I can convince you with a simple example, if you would like.~
== C-ARANB ~[ARAN] Aye, then, go ahead an' try.~
== SOLA ~[sOLAUFEIN] Have you ever known a Drow to communicate through poetry and song?~
== C-ARANB ~[ARAN] Point taken. An' I have to admit; mayhap I be wrong about th' only good Drow bein' a dead Drow. Mayhap there be one or two what be reasonable bein's.~
EXIT
Now comes the interesting part. As written above, Aran's initiated banter will play, but on the vast majority of installs, Solaufein's will not. Why? Well...
for one thing, Solaufein does not have a banter file.
Oh, he does - but he doesn't. Because Back In The Day™, in the very first mods, the same file was used for both banter and for joined file, and the NPC script switched things around.
And for a second thing, the RPG Solaufein Flirt pack adds a PID-style flirt to SOLA, so even if we call it, it will never play, as he flirts will superceed it in evaluation order. Actually, they might - the flirts are set on Player1 being the gabber, but if the folks there add the !Gabber(Player1) blocks I use, then they will have a wall - and someone, sometime, is going to build a proper PID for the lad. In a regular install, players will use the "obvious", and install Solaufein, then the RPG Flirt pack.
Two traditional ways to fix this - play around with weighting, moving the banter higher up in the order. Tricky, risks Weimer's work being blocked, and subject to install order problems. Yuck.
(actually, an additional one - start messing about with Solaufein's .bcs. Double-triple-yuck).
Or take the cheap way out - and don't have Solaufein actually start the banter - have it triggered from Aran's file. Since I have control of Aran's stuff, and not Solaufein's, I'm going "cheap, cheap, cheap", to quote Meilcamp!
Changed file entry -
/* SoA Banters: Solaufein<>Aran : cmorgan */
CHAIN IF ~Global("c-solaufein1aran","GLOBAL",0)
InParty(Myself) !StateCheck(Myself,CD_STATE_NOTVALID)
InParty("SOLA") InMyArea("SOLA") !StateCheck("SOLA",CD_STATE_NOTVALID)
CombatCounter(0) !See([ENEMY])~
THEN C-ARANB c-solaufein1arantalk
== C-ARANB ~[ARAN] Blighted stone. Got one in my boot, I do.~
DO ~SetGlobal("c-solaufein1aran","GLOBAL",1)~
== SOLA ~[sOLAUFEIN] Ah, my prejudiced fellow traveller - speech is a mirror of the soul. As a man speaks, so is he. I ask you now: who is Aran?~
== C-ARANB ~[ARAN] I don't rightly understand th' question. Wait... mayhap I do. Do you be sayin' th' way a man speaks be a good indication o' his personality, his habits, an' so forth?~
== SOLA ~[sOLAUFEIN] Why, yes. I think that is a fair assessment.~
== C-ARANB ~[ARAN] That be a right interestin' idea. But I don't think it applies to me. You want to know me, you look at how I write, an' ignore th' language.~
== SOLA ~[sOLAUFEIN] I do not think that will be a fair picture. With your scribing, you have time to think, to craft your ideas. In your speech, there is an immediacy that shows a raw truth.~
== C-ARANB ~[ARAN] Mayhap. Or mayhap we have to agree to disagree, eh?~
== SOLA ~[sOLAUFEIN] I can convince you with a simple example, if you would like.~
== C-ARANB ~[ARAN] Aye, then, go ahead an' try.~
== SOLA ~[sOLAUFEIN] Have you ever known a Drow to communicate through poetry and song?~
== C-ARANB ~[ARAN] Point taken. An' I have to admit; mayhap I be wrong about th' only good Drow bein' a dead Drow. Mayhap there be one or two what be reasonable bein's.~
EXIT
And that, as they say, is that

The Authorship stuff is always a PITA in this community, with a long history of oddities. But in general, I am using "authorized" to mean "the original author and I wrote up sketches and worked together", and "unauthorized" to cover "the original author did not participate". Many modders have said that they welcome additional crossmod content but are no longer in the community; others have said "no way I am not going to have anyone write lines for my NPC whether I am around or not". Bigger discussion at The Sensuous Permission-Free Crossmod Banter Author, By Mr. C .
To be completely honest, as I have said before in other posts, I'm with Nythrun on the silliness of "authorizing" and "unauthorizing" content built by unauthorized extrapolation of characters created for and registered to BioWare™. But I am not going to go out of my way and poke modders with definitive views that are different than my own, because (after all) hey, it's a game, and it isn't worth making someone all upset over writing a few lines of dialog. So, for instance, I know Theacefes is a person who prefers not to have other folks write lines for her character. I'll poke about and bug her, and with any luck we'll come up with some fun stuff, all "authorized", if she likes and has time. Why tick her off by writing stuff for Sarah? On the "unauthorized=not original modder" front, under my definition, Taisha and Kivan fall into the category even though each of their authors has given blanket permission for other folks to write banters for those characters. Domi has even worked up one - but the other, i went ahead and created something on my own that I thought she would like.
In this case, the whole discussion is moot, as I expect I can bribe/beg/plead with Domi to take a look and either say "sure" or give edits. So, Kivan SoA content. To be blunt, Kivan and Aran in ToB are not likely to have much to be chatting about; if something pops up, we can always add it.
Two banters in SoA
file: aranw_aware_kivan_SoAbanters.d
/* SoA Banters: Aran<>Kivan : Domi and cmorgan */
CHAIN IF ~Global("c-aran1kivan","GLOBAL",0)
InParty(Myself) !StateCheck(Myself,CD_STATE_NOTVALID)
InParty("p#kivan") InMyArea("p#kivan") !StateCheck("p#kivan",CD_STATE_NOTVALID)
CombatCounter(0) !See([ENEMY])~
THEN C-ARANB c-aran1kivantalk
~[ARAN] Blighted elves.~
DO ~SetGlobal("c-aran1kivan","GLOBAL",1)~
== BP#KIV ~[KIVAN] What?!~
== C-ARANB ~[ARAN] It be bad enough you live for several lifetimes.~
== BP#KIV ~[KIVAN] I have lost the ability to see my long life as a blessing, human.~
== C-ARANB ~[ARAN] Aye, an' add all that broodin' angst...~
== BP#KIV ~[KIVAN] How--~
== C-ARANB ~[ARAN] Hells, there be no chance for us fireflies wi' our own womenfolk, eh?~
== BP#KIV ~[KIVAN] Who can understand those humans?~
== C-ARANB ~[ARAN] Corellon's Sword, give it a shot, eh, or you might catch yourself some rough explainin'.~
EXIT
/* SoA Banters: Aran<>Kivan : cmorgan (don't blame Domi if it sucks, and when she gets time she might fix it up ) */
CHAIN IF ~Global("c-kivan1aran","GLOBAL",0)
InParty(Myself) !StateCheck(Myself,CD_STATE_NOTVALID)
InParty("c-aran") InMyArea("c-aran") !StateCheck("c-aran",CD_STATE_NOTVALID)
CombatCounter(0) !See([ENEMY])~
THEN BP#KIV c-kivan1arantalk
~[KIVAN] No, no... 'ah-MIN hi-RAE-tha'.~
DO ~SetGlobal("c-kivan1aran","GLOBAL",1)~
== C-ARANB ~[ARAN] 'ayahaMin hiRAYtah'.~
== BP#KIV ~[KIVAN] 'ah-MIN hi-RAE-tha'.~
== C-ARANB ~[ARAN] 'ayaha-MIN hi-RAYtah'.~
== BP#KIV ~[KIVAN] For the hundredth time, you must let the syllables sing. Gutteral harshness, amin uuma merna ta.~
== C-ARANB ~[ARAN] Blighted tongue. Serves me right for tryin' to speak a language what takes half a Xvim-kissed century to learn. It be a fair amount easier to write this than to speak it, eh?~
== BP#KIV ~[KIVAN] Perhaps in your case, that would be best.~
== C-ARANB ~[ARAN] Aye. What be th' chance I am actually goin' to say 'I'm sorry' in elven anyways, or any other language for that matter. Hey... how do you say 'take me to your bedchamber'?~
== BP#KIV ~[KIVAN] Shevarash, tua amin! Amin lava.~
EXIT
Installed and tested.

Finalized with Sister V, ready for inclusion with the AranW_Aware mod (if this post goes away due to forum stuff, no problems - it will be backed up at GitHub this weekend when the "aranw_aware" mod is cleared to be placed on the install after the current unreleased beta - my browser is showing "offline but I can still apparently post...)
When there is an active modder it is a blast to play "build the banters" with them. Way cool fun. And Sister V even voiced Angelo's stuff, which ups the ante for me...
Here are the ones Sister V and I came up with via PM, after suggestions in other threads:
Two banters for ToB
File: aranaware_angelo_ToBbanters.d
/* ToB Banters: Aran<>Angelo : Wine and Travel Story Swap : by Sister V and cmorgan */
CHAIN IF ~Global("c-aran3angelo","GLOBAL",0)
InParty("ADAngel") InMyArea("ADAngel") !StateCheck("ADAngel",CD_STATE_NOTVALID)
InParty(Myself) !StateCheck(Myself,CD_STATE_NOTVALID)
CombatCounter(0)
!See([ENEMY])
~ THEN C-ARN25B c-aran3angelotalk
~[ARAN] Cyric's Black Heart, this travelin' about builds up a powerful thirst. Give a lad a story, eh? Tell me what be th' best drink you ever did have.~
DO ~SetGlobal("c-aran3angelo","GLOBAL",1)~
== BADANG25 ~[ANGELO] Well, in Maztica, there is a concoction they call the Blood of the Serpent God. Refreshingly literal; it's a gods-damned viper stopped up in a bottle of ale. They say the trace of venom makes you see the face of Ao. Bought some of a peddler, and happen I saw Her face. Didn't you know She's a woman? Most beautiful lass you ever saw. And I understood why She made it all, you know; out of love, and I loved Her...the worst drink followed the best. On the third bottle, the face became my dam's, telling me what a disappointment I'd always been till the early morning hours. ~
== BADANG25 ~[ANGELO] That's how it goes. Anyway, what's yours?~
== C-ARN25B ~[ARAN] Well, th' best drink I ever did have were some right smart tastin' meade from up Moonshae ways. O' course, I didn't know it were stronger than most, an' I fell to drinkin' wi' a nice lass. She seemed to get prettier th' more I drank, an' th' more I drank she drank, an' soon enough we fell to sparkin'. It were one hells o' a night.~
== BADANG25 ~[ANGELO] Mayhap I sense a common theme. What of the morning to follow? ~
== C-ARN25B ~[ARAN] I done learned my lesson. I woke up wi' a Kossoth-kissed hangover, no coinpurse, an' somehow half o' my clothin' were missin' as well. That were a long walk back to barracks, that were. I never did know how cold a winter sea-breeze could be until I tried it wi' very little to protect me from th' waist down.~
== BADANG25 ~[ANGELO] Without the bitter aftertaste, I reckon, we'd not know the medicine worked. ~
== C-ARN25B ~[ARAN] Aye. That be a right powerful truth.~
EXIT
/* ToB Banters: Angelo<>Aran: heartsick */
CHAIN IF ~Global("c-angelo4aran","GLOBAL",0)
InParty(Myself) !StateCheck(Myself,CD_STATE_NOTVALID)
InParty("c-aran") InMyArea("c-aran") !StateCheck("c-aran",CD_STATE_NOTVALID)
CombatCounter(0)
!See([ENEMY])
~ THEN BADANG25 c-angelo4arantalk
~[ANGELO] Gods, all this killing makes me heartsick. Happen I think there's nothing more beautiful than a farmer maid singing to herself, hanging out the laundry as night falls... ~
DO ~SetGlobal("c-angelo3aran","GLOBAL",1)~
== C-ARN25B ~[ARAN] Now, that do sound right pretty. But on my account, swap in some red-tressed, nip-an'-tuck waisted, green-eyed slip o' a barmaid, an' I think I'd like that right proper.~
== BADANG25 ~[ANGELO] To each his own.~
== C-ARN25B ~[ARAN] Turnin' soft in your old age, eh? You done figured out what th' rest of us knew long ago. Welcome to th' human race.~
== BADANG25 ~[ANGELO] I may not join the rest of you yet. I'd still have the advantage of yon imagined farmer maid.~
== C-ARN25B ~[ARAN] Blighted sure, on any account. Not while she be singin', though, on account o' you won't want it to end. No tears shed on proper sparkin'.~
== BADANG25 ~[ANGELO] But while the song lasts, it is enough to make a man cry.~
== C-ARN25B ~[ARAN] But a man never cries, does he?~
== BADANG25 ~[ANGELO] Nay. Never.~
EXIT
Four banters for SoA
File: aranaware_angelo_SoAbanters.d
/* SoA Banters: Aran<>Angelo : Orders are Orders : by Sister V and cmorgan */
CHAIN IF ~Global("c-aran1angelo","GLOBAL",0) InParty("ADAngel") InMyArea("ADAngel") !StateCheck("ADAngel",CD_STATE_NOTVALID) InParty(Myself) !StateCheck(Myself,CD_STATE_NOTVALID) CombatCounter(0) !See([ENEMY])~ THEN C-ARANB c-aran1angelotalk
~[ARAN] So... you wanted to kill <PRO_HIMHER>?~
DO ~SetGlobal("c-aran1angelo","GLOBAL",1)~
== BADANGEL ~[ANGELO] Orders are orders.~
== C-ARANB ~[ARAN] True.~
== BADANGEL ~[ANGELO] Things were different.~
== C-ARANB ~[ARAN] Heh. I weren't there.~
== BADANGEL ~[ANGELO] Good thing. I would have taken you out first.~
== C-ARANB ~[ARAN] Scared o' me?~
== BADANGEL ~[ANGELO] Just can't stand your breath.~
== C-ARANB ~[ARAN] Yeah, screw you, too.~
== BADANGEL ~[ANGELO] Thanks, but I prefer women.~
== C-ARANB ~[ARAN] Heh.~
EXIT
/* SoA Banters: Angelo<>Aran : Manners : by Sister V and cmorgan */
CHAIN IF ~Global("c-angelo1aran","GLOBAL",0) InParty(Myself) !StateCheck(Myself,CD_STATE_NOTVALID) InParty("c-aran") InMyArea("c-aran") !StateCheck("c-aran",CD_STATE_NOTVALID) CombatCounter(0) !See([ENEMY])~ THEN BADANGEL c-angelo1arantalk
~[ANGELO] Aran, you curse like a Calimshite sailor. Show some respect. ~
DO ~SetGlobal("c-angelo1aran","GLOBAL",1)~
== C-ARANB ~[ARAN] Ha! This unwashed, whorin' drunkard thinks t' teach
me a lesson in manners? ~
== BADANGEL ~[ANGELO] A gentleman can whore and drink, and be a gentleman still. That's the Sembian way. ~
== C-ARANB ~[ARAN] Well, th' Sembian way be a mite bit strange. But it don't rightly matter, on account o' I never did be no gentleman.~
== BADANGEL ~[ANGELO] How incisive. ~
EXIT
/* SoA Banters: Aran<>Angelo : Using Drugs : by Sister V and cmorgan */
CHAIN IF ~Global("c-aran2angelo","GLOBAL",0)
InParty("ADAngel") InMyArea("ADAngel") !StateCheck("ADAngel",CD_STATE_NOTVALID)
InParty(Myself) !StateCheck(Myself,CD_STATE_NOTVALID)
CombatCounter(0) !See([ENEMY])~
THEN C-ARANB c-aran2angelotalk
~[ARAN] Usin' again?~
DO ~SetGlobal("c-aran2angelo","GLOBAL",1)~
== BADANGEL ~[ANGELO] Aye, happen I am.~
== C-ARANB ~[ARAN] You are a blighted Cyric-buggerin' bastard.~
== BADANGEL ~[ANGELO] And worse.~
== C-ARANB ~[ARAN] It's goin' to kill you. An' probably me. Mayhap <PRO_HIMHER>, too.~
== BADANGEL ~[ANGELO] Aye.~
== C-ARANB ~[ARAN] Goin' to be able to quit forever?~
== BADANGEL ~[ANGELO] Nay.~
== C-ARANB ~[ARAN] ... ~
== BADANGEL ~[ANGELO] But I'm not using today.~
== C-ARANB ~[ARAN] Good.~
== BADANGEL ~[ANGELO] Who are you, my father?~
== C-ARANB ~[ARAN] Nope. But for all I know, you are mine, you soddin' philanderin' bugger.~
== BADANGEL ~[ANGELO] Cheers to that thought.~
EXIT
/* SoA Banters: Angelo<>Aran : le soldat : by Sister V and cmorgan */
CHAIN IF ~Global("c-angelo2aran","GLOBAL",0) InParty(Myself) !StateCheck(Myself,CD_STATE_NOTVALID) InParty("c-aran") InMyArea("c-aran") !StateCheck("c-aran",CD_STATE_NOTVALID) CombatCounter(0) !See([ENEMY])~ THEN BADANGEL c-angelo2arantalk
~[ANGELO] Aran? I wanted to say, I appreciate having a soldier at my back. ~
DO ~SetGlobal("c-angelo2aran","GLOBAL",1)~
== C-ARANB ~[ARAN] Flattery? What do you be wantin' o' me now? ~
== BADANGEL ~[ANGELO] Why, nothing but your company. Be <CHARNAME>'s other companion's ever-so-talented, nothing beats old-fashioned army discipline.~
== C-ARANB ~[ARAN] An' adventurin' company beats th' blighted hells out o' army life any day, as far as food an' coin go.~
== BADANGEL ~[ANGELO] Don't I know it. Let's have a drink to that, eh? ~
EXIT

SoA: Two banters, plus one Underdark Banter
File: aranaware_saerileth_SoAbanters.d
/* SoA Banters : Aran-Saerileth */
CHAIN IF ~Global("c-aran1saerileth","GLOBAL",0)
InParty(Myself) !StateCheck(Myself,CD_STATE_NOTVALID)
InParty("Saerileth") InMyArea("Saerileth") !StateCheck("Saerileth",CD_STATE_NOTVALID)
CombatCounter(0) !See([ENEMY])~
THEN C-ARANB c-aran1saerilethbanter
~[ARAN] Lady Paladin, I'd be a might happier if you would take th' time to answer me a question, beggin' my impertinence.~
DO ~SetGlobal("c-aran1saerileth","GLOBAL",1)~
== BSAERILE ~[sAERILETH] Speak on, Aran.~
== C-ARANB IF ~!Global("SaerilethRomanceActive","GLOBAL",2)~ THEN ~[ARAN] <CHARNAME>'s rightly leader here, an' I expect has reason to have so young a lass on th' front lines. I was workin' trade costers m'self at your age, so I got no problem wi' that. But you seem a bit more focused on what be right an' wrong than what be practical.~
== C-ARANB IF ~Global("SaerilethRomanceActive","GLOBAL",2)~ THEN ~[ARAN] <CHARNAME>'s rightly leader here, an' I expect has reason to have so young a lass on th' front lines. I was workin' trade costers m'self at your age, so I got no problem wi' that. But you both seem a mite bit more than just comrades in arms.~
== BSAERILE IF ~!Global("SaerilethRomanceActive","GLOBAL",2)~ THEN ~[sAERILETH] Which would thou choose when the choice must be made: goodness or utility? There is but one answer.~
== BSAERILE IF ~Global("SaerilethRomanceActive","GLOBAL",2)~ THEN ~[sAERILETH] I am his love, and he is mine own.~
== C-ARANB ~[ARAN] An' this answer serves Tyr, on account o' you bein' his Chosen an' all?~
== BSAERILE IF ~!Global("SaerilethRomanceActive","GLOBAL",2)~ THEN ~[sAERILETH] My life is, and hath ever been, Tyr's only. I shall follow his will so long as I have breath.~
== BSAERILE IF ~Global("SaerilethRomanceActive","GLOBAL",2)~ THEN ~[sAERILETH] My life is, and hath ever been, Tyr's. I shall follow his will so long as I have breath. <CHARNAME> knows this. He will understand the depth of my faith.~
== C-ARANB ~[ARAN] I'm not sure I understand that kind o' faith, is all. I never did.~
EXIT
/* SoA Banters : Saerileth-Aran */
CHAIN IF ~Global("c-saerileth1aran","GLOBAL",0)
InParty(Myself) !StateCheck(Myself,CD_STATE_NOTVALID)
InParty("c-aran") InMyArea("c-aran") !StateCheck("c-aran",CD_STATE_NOTVALID)
CombatCounter(0) !See([ENEMY])~
THEN BSAERILE c-saerileth1aranbanter
~[sAERILETH] Why dost thou reject the gods? The foul-mouthed oaths thou dost speak do disservice to all.~
DO ~SetGlobal("c-saerileth1aran","GLOBAL",1)~
== C-ARANB ~[ARAN] Come, now, Saerileth. There must o' been a time when you were not so tight-ar... err... no, I suppose you never did let your hair down, so to speak. You haven't had enough time in life yet.~
== BSAERILE ~[sAERILETH] Thy jesting is misplaced, and I am not amused.~
== C-ARANB ~[ARAN] Now, don't you be gettin' all offended, lass. There be a place for a harmless bit o' language, sometimes. An' Tymora seems to be fine wi' me playin' fast an' loose.~
== C-ARANB IF ~OR(2) Class("c-aran",PALADIN_ALL) Class("c-aran",CLERIC_ALL)~ THEN ~[ARAN] How about we not get our respective gods into no minor squabble just on account o' my blasphemin', eh?~
== C-ARANB IF ~!Class("c-aran",PALADIN_ALL) !Class("c-aran",CLERIC_ALL)~ THEN ~[ARAN] I know I am no champion o' virtue. I done lost that in a card game when I... never you mind that. But I try to back <CHARNAME>, an' you back <CHARNAME>, so that has to count for somethin', eh?~
== BSAERILE IF ~OR(2) Class("c-aran",PALADIN_ALL) Class("c-aran",CLERIC_ALL)~ THEN ~[sAERILETH] I do pity thee that thou knowest not yet the wishes of thy god. Foul-mouthed blasphemy will forfeit the good will of all of the gods.~
== BSAERILE IF ~!Class("c-aran",PALADIN_ALL) !Class("c-aran",CLERIC_ALL)~ THEN ~[sAERILETH] I do pity thee that thou knowest not yet the power of worshiping one god above all others. Foul-mouthed blasphemy will forfeit the good will of all of the gods.~
== C-ARANB ~[ARAN] Is this somethin' where you are likely to raise a sword against me if I curse a mite here an' there?~
== BSAERILE ~[sAERILETH] As long as thou dost tread the path of good, I will sojourn with thee. But I warn thee... curb thy language.~
== C-ARANB ~[ARAN] You know, hearin' those words from a little slip o' a girl not even out o' apprentice age might be laughable in some circumstances. But seein' as you can hold your own in combat an' healin', I will take you right serious. At least, as far as I can. Hard to break old habits.~
EXIT
/* SoA UD Banters : Aran-Saerileth */
CHAIN IF ~Global("c-drow","GLOBAL",1) Global("c-aranudsaerileth","GLOBAL",0)
InParty(Myself) !StateCheck(Myself,CD_STATE_NOTVALID)
InParty("Saerileth") InMyArea("Saerileth") !StateCheck("Saerileth",CD_STATE_NOTVALID)
CombatCounter(0) !See([ENEMY])~
THEN C-ARANB AranSaerUDBanter
~[ARAN] I know I don't rightly take you serious all th' time. You be a fine companion, though a mite bit stubborn an' mayhap a bit fanatical. But here, I can see why a lad might follow in your footsteps a mite. This place deserves to be eradicated.~
DO ~SetGlobal("c-aranudsaerileth","GLOBAL",1)~
== BSAERILE ~[sAERILETH] 'Twould be false in me to deny that the continued existence of this place doth trouble me.~
== C-ARANB ~[ARAN] I'd think it be time to call forth a holy crusade, an' clear 'em off out o' here. Mayhap destroy evey last one o' them.~
== BSAERILE ~[sAERILETH] Our mortal lives do oft seem full of injustice, but if our lots are here unjust, we may rest assured that the Maimed Tyr will not deny us justice when he doth judge us. But all of these creatures have chosen to serve evil. Their judgment hath been wrought, their lot decided, and eventually they shall fall to the power of Tyr.~
== C-ARANB ~[ARAN] I don't suppose there be some who might follow a good cause down here, eh? Mayhap one or two redeemed souls, trapped alongside all this scurryin' evil?~
== BSAERILE ~[sAERILETH] Forgive me. My zeal oft overcomes my discretion. In this matter, I spoke hastily; Tyr is the Judge, and I am his Champion. Perhaps there are a few souls the Maimed One would have me spare. It is not my decision to make.~
== C-ARANB ~[ARAN] Well, now you done got me worried. Usually, you be th' one takin' things to extremes. I was about to suggest we just send 'em all Tyr's way, an' let him figure out who goes where. I think down here, I be a might bit more comfortable wi' "guilt by association".~
EXIT
SoA: comment on romance
File: aranaware_saerileth_SoAcomment.baf
/* Saerileth Commentary SoA : Are you sure about this romance? : Activate */
IF
InParty("Saerileth")
InMyArea("Saerileth")
!StateCheck("Saerileth",CD_STATE_NOTVALID)
InParty(Myself)
!StateCheck(Myself,CD_STATE_NOTVALID)
CombatCounter(0)
Global("SaerilethRomanceActive","GLOBAL",2)
Global("c-aranpcsaerrom","GLOBAL",0)
THEN
RESPONSE #100
SetGlobal("c-aranpcsaerrom","GLOBAL",1)
END
/* Saerileth Commentary SoA : Are you sure about this romance? : Initiate */
IF
GlobalTimerExpired("c-aransmalltalk","GLOBAL") // Timer (game-time) has expired
Global("c-silencearan","GLOBAL",0) // a PID - set "silencer" for folks who hate unsolicited comments
Global("c-aranpcsaerrom","GLOBAL",1)
InParty(Myself)
!StateCheck(Myself,CD_STATE_NOTVALID)
!StateCheck(Player1,CD_STATE_NOTVALID)
THEN
RESPONSE #100
StartDialogNoSet(Player1)
END
File: aranaware_saerileth_SoAdialog.d
APPEND C-ARANJ
/* Saerileth Commentary SoA : Are you sure about this romance? : Dialog */
IF WEIGHT #-1 ~Global("c-aranpcsaerrom","GLOBAL",1)~ THEN BEGIN carsar1
SAY ~[ARAN] Now, this be naught in th' way o' my business. But I have to ask... are you an' Saerileth gettin' serious?~
++ ~[PC] Of course not. That is a ridiculous suggestion.~ DO ~SetGlobal("c-aranpcsaerrom","GLOBAL",2)~ + carsar2
++ ~[PC] What do you think?~ DO ~SetGlobal("c-aranpcsaerrom","GLOBAL",2)~ + carsar3
++ ~[PC] Yes.~ DO ~SetGlobal("c-aranpcsaerrom","GLOBAL",2)~ + carsar4
++ ~[PC] Are we holding hands and kissing occasionally? Yes, I think we are. Are we getting serious? I am not sure.~ DO ~SetGlobal("c-aranpcsaerrom","GLOBAL",2)~ + carsar4
++ ~[PC] Do I ask you about your dalliances with half of Faerun?~ DO ~SetGlobal("c-aranpcsaerrom","GLOBAL",2)~ + carsar3
++ ~[PC] Speak of this to me again, and you will be looking for a new line of work. And perhaps searching for your missing head.~ + carsar10
END
IF ~~ carsar2
SAY ~[ARAN] You must be blighted blind, then. Th' lass looks at you wi' th' intensity o' first love, an' th' passions o' a full grown woman. You'd best be careful around her, on account o' she might get th' wrong idea - or worse, th' right one.~
++ ~[PC] I don't really need to be careful. I think I am falling for her.~ + carsar4
++ ~[PC] Are we holding hands and kissing occasionally? Yes, I think we are. Are we getting serious? I am not sure.~+ carsar8
++ ~[PC] Do I ask you about your dalliances with half of Faerun?~ DO ~SetGlobal("c-aranpcsaerrom","GLOBAL",2)~ + carsar3
++ ~[PC] Speak of this to me again, and you will be looking for a new line of work. And perhaps searching for your missing head.~ + carsar10
END
IF ~~ carsar3
SAY ~[ARAN] Look, <CHARNAME>, I love th' way women move, th' way they speak, th' way they smell, th' way they smile... hells, I just love women. But me an' someone what wears th' flower be a mite different than what you are talkin' about. First off, we all have th' age o' consent. Second off, if I tick off th' lady in question, she may box my ears, kick me, get me thrown into th' street, or mayhap carve me up a little, as be her choice. With Saerileth...~
IF ~~ THEN GOTO carsar8
END
IF ~~ carsar4
SAY ~[ARAN] Do you be kiddin' wi' me? A lass what has seen less than sixteen winters? Give her a year to come o' age a' make her own chices, an' mayhap I wouldn't kick her out o' my b... err... well, she is pretty an' all, I guess. But she don't know squat about livin', let alone havin' a serious relationship. Tyr's Blasted Buttocks, don't send a child to do adult's work, <CHARNAME>. Love isn't just for play.~
++ ~[PC] Take one look into her eyes, and tell me that she does not love me.~ + carsar5
++ ~[PC] She is the Chosen of Tyr. She is strong enough, wise enough, and driven enough to know her own mind. Do you really think we would be together if she was not set on having it that way?~ + carsar6
++ ~[PC] What, are you jealous? I did not know you were in love with her too.~ + carsar7
++ ~[PC] It is a passing crush, and I intend to have as much fun with her as her religious views will allow.~ + carsar8
++ ~[PC] So now you are an expert on serious relationships? That is a laugh. You don't even remember the barmaid's name the next morning.~ + carsar3
END
IF ~~ carsar5
SAY ~[ARAN] Aye, she does. That be even more dangerous.~
IF ~~ THEN GOTO carsar9
END
IF ~~ carsar6
SAY ~[ARAN] Mayhap you are right. Mayhap she be older than her years, an' wiser than a sage; mayhap she be mature enough to understand that relationships be work as well as fun. But all that wisdom an' such be packed into a body what be still growin' an' changin'.~
IF ~~ THEN GOTO carsar9
END
IF ~~ carsar7
SAY ~[ARAN] Jealous? No possibility o' that, boyo. She scares me half to death.~
IF ~~ THEN GOTO carsar9
END
IF ~~ carsar9
SAY ~[ARAN] Look, <CHARNAME>, I love th' way women move, th' way they speak, th' way they smell, th' way they smile... hells, I just love women. But me an' someone what wears th' flower be a mite different than what you are talkin' about. First off, we all have th' age o' consent, meanin' we can make our own choices. Second off, if I tick off th' lady in question, she may box my ears, kick me, get me thrown into th' street, or mayhap carve me up a little, as be her choice. With Saerileth...~
IF ~~ THEN GOTO carsar8
END
IF ~~ carsar8
SAY ~[ARAN] You know how 'hell hath no fury like a woman scorned'? Well, take one look at her an' all you see is slim little hips, curves to die for, an' lips beggin' to be kissed - not one bit o' th' hells about her. Take a look in her eyes, though, an' you see an intensity an' a devotion that be downright scary. Now picture her if you hurt her. There won't be any place on any o' th' Planes where you could hide from her wrath. She'll rip you in half wi' her bare hands, then heal you up just so she can do it again.~
++ ~[PC] You are taking this too seriously. I am much more powerful than she is. She will never be able to defeat me.~ + carsar10
++ ~[PC] I don't intend to hurt her. I think she may be my soulmate.~ + carsar10
++ ~[PC] She's a bunny rabbit. What's she going to do, nibble my bum?~ + carsar10 /* heh - Monty Python is fun. But this one might be a good one to lose. */
++ ~[PC] This is my business, not yours. Keep out of it.~ + carsar10
END
IF ~~ carsar10
SAY ~[ARAN] None o' my business, anyways. You think you can handle it. But I have been on th' receivin' end o' a regular woman what feels she has been wronged, an' it were a mite painful. Breakin' th' heart o' someone with th' feelin's o' adolescent first love who just happens to be an accomplished fighter an' th Chosen o' Tyr... that's bloody suicide.~
IF ~~ THEN EXIT
END
END
Installed and tested.

We will insert the GitHub link once it is up right <<here>>, but for now, let's take inventory and start setting up for the documentation.
Material written and integrated so far (except for the Gavin crossmod stuff that Berelinde and I are still tinkering with):
Divine Remix
aranw_aware will open pathways that recognize if your PC is using one of the Cleric Kits from Divine Remix.
Kits Recognized:
Heartwarder of Sune
Feywarden of Correlon
Strifeleader of Cyric
Painbearer of Ilmater
Lorekeeper of Oghma
Firewalker of Kossuth
Silverstar of Selune
Holy Strategist of the Red Knight
Nightcloak of Shar
Battleguard of Tempus
Authlim of Iyachtu Xvim
NPC Mod Banters
aranw_aware will add banters between Aran and the NPC in question if both Aran Whitehand and the other NPC mod are installed before running aranw_aware.
NPC Mods Recognized:
Laufey's Edwin Romance (one interjection in a conflict between Edwin and Anomen) "unauthorized"
Angelo (two banters in SoA, two in ToB) "authorized", by Sister Vigilante and cmorgan; sound clips by Sister Vigilante
Kivan (two banters in SoA) - "authorized" and "unauthorized" one by Domi and cmorgan, another by cmorgan (and hopefully she will ok it when she has the chance).
Saerileth (three banters and one comment in SoA) "unauthorized", by cmorgan - if they stink, blame me and not Netharin/Sillara
Tashia Remix, by Michael "Arian" Lyashenko, Bri, and Lord Ernie (two banters in SoA, two in ToB) "unauthorized", by cmorgan - if they stink, blame me and not Bri, Lord Ernie, or the original author, Michael "Arian" Lyashenko
Tsujatha NPC MOD For Baldur’s Gate 2 (TOB Required) by Sillara of the Tamari (two banters in SoA) "unauthorized", by cmorgan - if they stink, blame me and not Sillara
Solaufein Romance Mod Pack, by Westley Weimer (two banters in SoA) "unauthorized", by cmorgan - if they stink, blame me and not Westley Weimer
Rather than filling the forum up with code, I will link the materials to github, so folks can see the .tp2 materials that way.